

```
In[*]:= ClearAll["Global`*"]
```

- Preliminaries
- We allow for arbitrary  $\mu$  and  $\alpha$
- Throughout, we operate with interior parameter values  $\mu, \alpha \in (0,1)$ . Including the boundary cases is straightforward.
- We set  $\delta = 1$  for ease of exposition.

```
In[*]:= (* Initialize Utilities *)
```

```
In[*]:= ulow = u;
```

```
In[*]:= uhigh = ulow + Δ;
```

```
In[*]:= (* Expected Utility *)
```

```
In[*]:= expout = ulow + μ (uhigh - ulow)
```

```
Out[*]:= u + Δ μ
```

```
In[*]:= (* Predicted Expected Utility of Motivated biased Consumers *)
```

```
In[*]:= expmot = FullSimplify[μ uhigh + (1 - μ) (α uhigh + (1 - α) ulow)]
```

```
Out[*]:= u + Δ (α + μ - α μ)
```

```
In[*]:= (* Predicted Expected Utility of Unmotivated biased Consumers *)
```

```
In[*]:= expunmot = FullSimplify[μ (α ulow + (1 - α) uhigh) + (1 - μ) ulow]
```

```
Out[*]:= u - (-1 + α) Δ μ
```

```
In[*]:= assumptions = 1 > α > 0 && 1 > μ > 0 && u > 0 && Δ > 0 && c ∈ ℝ;
```

- Baseline Model -- Preparation for Welfare & Demand

```
In[*]:= (* Initiate Demand by motivated consumers, motd, and unmotivated consumers unmotd *)
```

```
In[*]:= (* Assume that consumers located at point x suffer "transport costs" equal to x;  
on any line of length L > 0 there is a mass L of consumers *)
```

```
In[*]:= (* Get cutoff consumers of each group - motivated and unmotivated *)
```

```
In[*]:= xmot = x /. Solve[expmot - p - x == 0, x][[1]];
xunmot = x /. Solve[expunmot - p - x == 0, x][[1]];

```

```
In[*]:= (* Demand follows from cutoff consumer multiplied by proportion of relevant group *)
```

```
In[*]:= (* motivated consumers: μ *)
```

```
In[*]:= motd = FullSimplify[μ * xmot]
```

```
Out[*]:= μ (-p + u + Δ (α + μ - α μ))
```

```
In[*]:= (* unmotivated consumers: (1-μ) *)
```

```
In[*]:= unmotd = FullSimplify[(1 - μ) * xunmot]
```

```
Out[*]:= (-1 + μ) (p - u + (-1 + α) Δ μ)
```

```
In[*]:= (* For Welfare we require the total transport costs paid by  
consumers of each type which again depend on the cutoff consumer *)
```

*In[ ]:=* (\* Transport Costs are equal to mass of consumer type  
time times integral of transport costs up to cutoff consumer \*)

*In[ ]:=* mottc =  $\mu$  Integrate[x, {x, 0, xmot}]

$$\text{Out[ ]:= } \frac{1}{2} \mu (-p + u + \alpha \Delta + \Delta \mu - \alpha \Delta \mu)^2$$

*In[ ]:=* unmottc =  $(1 - \mu)$  Integrate[x, {x, 0, xunmot}]

$$\text{Out[ ]:= } \frac{1}{2} (1 - \mu) (-p + u + \Delta \mu - \alpha \Delta \mu)^2$$

*In[ ]:=* (\* Consumer Welfare of  
motivated: Demand times (actual exput minus price) net of total transport costs \*)

*In[ ]:=* motcw = FullSimplify[motd \* (exput - p) - mottc, assumptions]

$$\text{Out[ ]:= } -\frac{1}{2} \mu (p - u - \alpha \Delta + (-1 + \alpha) \Delta \mu) (-p + u + \Delta (\alpha (-1 + \mu) + \mu))$$

*In[ ]:=* (\* Same for unmotivated consumers \*)

*In[ ]:=* unmotcw = FullSimplify[unmotd \* (exput - p) - unmottc, assumptions]

$$\text{Out[ ]:= } \frac{1}{2} (-1 + \mu) (p - u + (-1 + \alpha) \Delta \mu) (-p + u + (1 + \alpha) \Delta \mu)$$

#### ■ Strategy 1: Full Targeting

*In[ ]:=* (\* Full targeting caters to some  
unmotivated consumers and hence requires  $p < \text{expunmot}$  \*)

*In[ ]:=* (\* Demand is then equal to the demand of both groups \*)

*In[ ]:=* dfull = FullSimplify[motd + unmotd]

$$\text{Out[ ]:= } -p + u + \Delta \mu$$

*In[ ]:=* (\* Solve Maximization Problem conditional on targeting strategy \*)

*In[ ]:=* tempsol = FullSimplify[Maximize[(p - c) \* dfull, p]]

$$\text{Out[ ]:= } \left\{ \frac{1}{4} (-c + u + \Delta \mu)^2, \left\{ p \rightarrow \frac{1}{2} (c + u + \Delta \mu) \right\} \right\}$$

*In[ ]:=* (\* Use solution to characterize profits, price, quantity, CW \*)

*In[ ]:=* proffull = tempsol[[1]];

pfull = p /. tempsol[[2]];

qfull = FullSimplify[dfull /. p → pfull];

cwfull = FullSimplify[motcw + unmotcw /. p → pfull, {1 >  $\mu$  > 0,  $\Delta$  > 0, 1 >  $\alpha$  > 0}];

*In[ ]:=* (\* Print Solutions \*)

*In[ ]:=* FullSimplify[{qfull, cwfull}, assumptions]

$$\text{Out[ ]:= } \left\{ \frac{1}{2} (-c + u + \Delta \mu), \frac{1}{8} (c^2 + u^2 + 2 u \Delta \mu + \Delta^2 \mu (4 \alpha^2 (-1 + \mu) + \mu) - 2 c (u + \Delta \mu)) \right\}$$

*In[ ]:=* (\* Obtain cost threshold such that solution is interior,  
i.e. such that price equals maximal price such that both consumer groups purchase \*)

```
In[ ]:= Solve[pfull == expunmot, c]
```

```
Out[ ]:= { {c → u + Δ μ - 2 α Δ μ} }
```

```
In[ ]:= (* Unique cutoff cost *)
```

```
In[ ]:= cfullhigh = FullSimplify[c /. Solve[pfull == expunmot, c][[1]]];
```

■ Strategy 2: Only Motivated

```
In[ ]:= (* This strategy requires p > expunmot,  
but p < expmot so that some motivated consumers purchase *)
```

```
In[ ]:= (* Demand in this case is composed only of the group of motivated consumers *)
```

```
In[ ]:= dexcl = motd
```

```
Out[ ]:= μ (-p + u + Δ (α + μ - α μ))
```

```
In[ ]:= (* Solve Maximization Problem conditional on Targeting Strategy *)
```

```
In[ ]:= tempsol = FullSimplify[Maximize[(p - c) * dexcl, p], assumptions]
```

```
Out[ ]:= { 1/4 μ (-c + u + Δ (α + μ - α μ))^2, {p → 1/2 (c + u + Δ (α + μ - α μ))} }
```

```
In[ ]:= (* Use solution to characterize profits, price, quantity, CW *)
```

```
In[ ]:= profexcl = tempsol[[1]];
pexcl = p /. tempsol[[2]];
qexcl = FullSimplify[dexcl /. p → pexcl];
cwexcl = FullSimplify[motcw /. p → pexcl];
```

```
In[ ]:= (* Print solutions *)
```

```
In[ ]:= FullSimplify[{qexcl, cwexcl}, assumptions]
```

```
Out[ ]:= { 1/2 μ (-c + u + Δ (α + μ - α μ)), -1/8 μ (c - u - α Δ + (-1 + α) Δ μ) (-c + u + Δ (3 α (-1 + μ) + μ)) }
```

```
In[ ]:= (* Get cost threshold such that solution is interior,  
i.e. price such that only motivated biased consumers purchase *)
```

```
In[ ]:= Solve[pexcl == expunmot, c]
```

```
Out[ ]:= { {c → u - α Δ + Δ μ - α Δ μ} }
```

```
In[ ]:= cexcllow = c /. FullSimplify[Solve[pexcl == expunmot, c]][[1]];
```

■ Characterization of Optimal Pricing -- derive cost threshold

■ Comparison of Full Targeting and Exclusive Targeting

```
In[ ]:= FullSimplify[Solve[profexcl == proffull, c], assumptions]
```

```
Out[ ]:= { {c → u + Δ (μ - α (√μ + μ))}, {c → u + Δ (α (√μ - μ) + μ)} }
```

```
In[ ]:= c1temp = c /. FullSimplify[Solve[profexcl == proffull, c], assumptions][[1]];
c2temp = c /. FullSimplify[Solve[profexcl == proffull, c], assumptions][[2]];
```

```
In[ ]:= FullSimplify[c1temp < exput < c2temp, assumptions]
```

```
Out[ ]:= True
```

```
In[ ]:= (* The relevant cutoff must be the lower root, i.e. the first solution *)
```

```
In[*]:= ctilde = c1temp;
```

- Verify that cutoff leads to interior solutions

```
In[*]:= FullSimplify[cfullhigh > ctilde, assumptions]
```

```
Out[*]:= True
```

```
In[*]:= FullSimplify[ctilde > cexcllow, assumptions]
```

```
Out[*]:= True
```

```
In[*]:= (* Interior solution ensured, pricing determined *)
```

- Cooling Off

```
■ =====
```

- Repeat same exercise as previously, only adjusting for different weights of the groups

```
In[*]:= (* Get cutoff consumers of each group - motivated and unmotivated *)
```

```
In[*]:= xmotco = x /. Solve[expmot - p - x == 0, x][[1]];
xunmotco = x /. Solve[expunmot - p - x == 0, x][[1]];
```

```
In[*]:= (* Use cutoff consumers to determine
demand from each group by accounting for weight *)
(* Demand by motivated consumers is that of twice motivated consumers *)
(* Demand by unmotivated consumers here
comprises those unmotivated in at least one period *)
```

```
In[*]:= motdco =  $\mu^2$  xmotco;
unmotdco =  $((1 - \mu) + \mu(1 - \mu))$  xunmotco;
```

```
In[*]:= (* For Welfare we require the total transport costs
paid by each type which again depend on the cutoff consumer *)
```

```
In[*]:= mottcco =  $\mu^2$  Integrate[x, {x, 0, xmotco}];
unmottcco =  $(\mu(1 - \mu) + (1 - \mu))$  Integrate[x, {x, 0, xunmotco}];
```

```
In[*]:= (* CW evaluated using actual expected utility exput *)
```

```
In[*]:= motcwco = FullSimplify[motdco (exput - p) - mottcco];
unmotcwco = FullSimplify[unmotdco * (exput - p) - unmottcco];
```

- Strategy 1: Exclusive Targeting,  $p > \text{expunmot}$

```
In[*]:= dexclco = FullSimplify[motdco];
```

```
In[*]:= (* Print for Demand Function *)
```

```
In[*]:= FullSimplify[dexclco]
```

```
Out[*]:=  $\mu^2 (-p + u + \Delta (\alpha + \mu - \alpha \mu))$ 
```

```
In[*]:= (* Solve Maximization Problem conditional on Targeting Strategy *)
```

```
In[*]:= tempsol = FullSimplify[Maximize[(p - c) * dexclco, p], assumptions]
```

```
Out[*]:=  $\left\{ \frac{1}{4} \mu^2 (-c + u + \Delta (\alpha + \mu - \alpha \mu))^2, \left\{ p \rightarrow \frac{1}{2} (c + u + \Delta (\alpha + \mu - \alpha \mu)) \right\} \right\}$ 
```

```
In[*]:= (* Use solution to characterize profits, price, quantity, CW *)
```

```

In[*]:= profexclco = tempsol[[1]];
pexclco = p /. tempsol[[2]];
qexclco = FullSimplify[dexclco /. p → pexclco];
cwexclco = FullSimplify[motcwco /. p → pexclco];

In[*]:= (* Print Solutions *)

In[*]:= FullSimplify[{qexclco, cwexclco}, assumptions]
Out[*]:=  $\left\{ \frac{1}{2} \mu^2 (-c + u + \Delta (\alpha + \mu - \alpha \mu)), -\frac{1}{8} \mu^2 (c - u - \alpha \Delta + (-1 + \alpha) \Delta \mu) (-c + u + \Delta (3 \alpha (-1 + \mu) + \mu)) \right\}$ 

In[*]:= (* Get cost thresholds so that solution is interior *)

In[*]:= Solve[pexclco == expunmot, c]
Out[*]:=  $\{ \{ c \rightarrow u - \alpha \Delta + \Delta \mu - \alpha \Delta \mu \} \}$ 

In[*]:= cexcllowco = c /. FullSimplify[Solve[pexclco == expunmot, c]][[1]];
  ■ Strategy 2: Full Targeting

In[*]:= dfullco = FullSimplify[motdco + unmotdco];

In[*]:= (* Print for Demand Function *)

In[*]:= FullSimplify[dfullco]
Out[*]:=  $-p + u + \Delta (1 + \alpha (-1 + \mu)) \mu$ 

In[*]:= (* Solve Maximization Problem conditional on Targeting Strategy *)

In[*]:= tempsol = FullSimplify[Maximize[(p - c) * dfullco, p], assumptions]
Out[*]:=  $\left\{ \frac{1}{4} (c - u + \Delta \mu (-1 + \alpha - \alpha \mu))^2, \{ p \rightarrow \frac{1}{2} (c + u + \Delta (1 + \alpha (-1 + \mu)) \mu) \} \right\}$ 

In[*]:= (* Use solution to characterize profits, price, quantity, CW *)

In[*]:= proffullco = tempsol[[1]];
pfullco = p /. tempsol[[2]];
qfullco = FullSimplify[dfullco /. p → pfullco];
cwfullco = FullSimplify[motcwco + unmotcwco /. p → pfullco, assumptions];

In[*]:= (* Print Solutions *)

In[*]:= FullSimplify[{qfullco, cwfullco}, assumptions]
Out[*]:=  $\left\{ \frac{1}{2} (-c + u + \Delta (1 + \alpha (-1 + \mu)) \mu), \frac{1}{8} (c^2 + u^2 + 2 u \Delta \mu (1 + \alpha - \alpha \mu) - 2 c (u + \Delta \mu (1 + \alpha - \alpha \mu)) + \Delta^2 \mu^2 (1 + \alpha (-1 + \mu) (-2 + \alpha (7 + \mu))) \right\}$ 

In[*]:= (* Show that cwfullco equates to expression in PDF *)

In[*]:= FullSimplify[cwfullco ==
  1/8 ((u - c) (u + 2 (1 + (1 - μ) α) μ Δ - c) + (1 + (1 - μ) α (2 - (7 + μ) α)) μ^2 Δ^2)]
Out[*]:= True

In[*]:= (* Get cost threshold so that solution is interior *)

In[*]:= Solve[pfullco == expunmot, c]
Out[*]:=  $\{ \{ c \rightarrow u + \Delta \mu - \alpha \Delta \mu - \alpha \Delta \mu^2 \} \}$ 

```

```
In[*]:= cfullhighco = c /. FullSimplify[Solve[pfullco == expunmot, c]][[1]];
```

- Characterize Optimal Targeting
- Comparison of Full and Exclusive Targeting

```
In[*]:= FullSimplify[Solve[profexclco == proffullco, c], assumptions]
```

```
Out[*]:= {{c -> u + Δ μ}, {c -> u + (1 - 2 α) Δ μ}}
```

```
In[*]:= c1temp = c /. FullSimplify[Solve[profexclco == proffullco, c], assumptions][[1]];
c2temp = c /. FullSimplify[Solve[profexclco == proffullco, c], assumptions][[2]];
```

```
In[*]:= FullSimplify[c1temp > c2temp, assumptions]
```

```
Out[*]:= True
```

```
In[*]:= (* The relevant cutoff must be the lower root, i.e. the second solution *)
```

```
In[*]:= ctildeco = c2temp;
```

```
In[*]:= (* Verify that this leads to pricing being interior *)
```

```
In[*]:= FullSimplify[cfullhighco > ctildeco, assumptions]
```

```
Out[*]:= True
```

```
In[*]:= FullSimplify[ctildeco > cexcllowco, assumptions]
```

```
Out[*]:= True
```

```
In[*]:= (* Interior targeting ensured, so pricing is determined *)
```

- Return Policy

```
In[*]:= (* Demand *)
```

```
In[*]:= (* First initialize predicted utility in motivated state for unmotivated consumers *)
```

```
In[*]:= uunmotmot = FullSimplify[α ulow + (1 - α) uhigh]
```

```
Out[*]:= u + Δ - α Δ
```

```
In[*]:= (* Initiate Demand -
tendering decisions depend on first period predicted utility for high state,
confirmation decisions potentially on utility in low state *)
```

```
In[*]:= (* xexclrp: Cutoff such that motivated consumers do not return,
xlowrp: Cutoff such that unmotivated consumers do not return *)
```

```
In[*]:= (* xmedrp: Cutoff such that unmotivated consumers tender in the first period *)
```

```
In[*]:= xexclrp = x /. Solve[uhigh - p - x == 0, x][[1]];
xmedrp = x /. Solve[uunmotmot - p - x == 0, x][[1]];
xlowrp = x /. Solve[ulow - x - p == 0, x][[1]];
```

```
In[*]:= (* Three strategies catering to different people *)
```

```

In[*]:= (* Consider first second-period motivated people who in the
         first period correctly predict the utility in the motivated state *)
         (* These are consumers who are motivated in the first period *)
         (* These consumers initially buy provided that  $u_{high} - p - x \geq 0$ 
            where  $x$  is their distance *)
         (* We can characterize the demand of these consumers
            and associated transport costs and CW *)

In[*]:= excludrp = FullSimplify[ $\mu^2 x_{exclrp}$ ];
         excltcpr =  $\mu^2 \text{Integrate}[x, \{x, 0, x_{exclrp}\}]$ ;
         exclcwrp = FullSimplify[ $excludrp * (u_{high} - p) - excltcpr$ ];

In[*]:= (* A second group of consumers are biased and initially unmotivated,
         but motivated in the second period at the return decision *)
         (* These consumers initially buy provided that  $u_{unmotmot} - p - x \geq 0$ 
            where  $x$  is their distance *)
         (* We can characterize the demand of these consumers
            and associated transport costs and CW *)
         (* This group is only relevant for  $p < u_{unmotmot}$  *)

In[*]:= meddrp = FullSimplify[ $(1 - \mu) \mu x_{medrp}$ ];
         medtcpr = FullSimplify[ $\mu (1 - \mu) \text{Integrate}[x, \{x, 0, x_{medrp}\}]$ ];
         medcwrp = FullSimplify[ $meddrp * (u_{high} - p) - medtcpr$ ];

In[*]:= (* The final group of consumers are the second-period unmotivated consumers *)
         (* Clearly, if  $p$  is low enough that they confirm the purchase,
            they also initially registered *)
         (* This group is only relevant for  $p < u_{low}$  *)

In[*]:= lowdrp =  $(1 - \mu) x_{lowrp}$ ;
         lowtcpr = FullSimplify[ $(1 - \mu) \text{Integrate}[x, \{x, 0, x_{lowrp}\}]$ ];
         lowcwrp = FullSimplify[ $lowdrp * (u_{low} - p) - lowtcpr$ ];

■ Get Demands for Different Strategies

In[*]:= (*  $p > u_{unmotmot}$  *)

In[*]:= dexclrp = FullSimplify[excludrp]

Out[*]:=  $(-p + u + \Delta) \mu^2$ 

In[*]:= (*  $u_{unmotmot} > p > u_{low}$  *)

In[*]:= dmedrp = FullSimplify[excludrp + meddrp]

Out[*]:=  $(-p + u + \Delta + \alpha \Delta (-1 + \mu)) \mu$ 

In[*]:= (*  $p < u_{low}$  *)

In[*]:= dfullrp = FullSimplify[excludrp + meddrp + lowdrp]

Out[*]:=  $-p + u + \Delta (1 + \alpha (-1 + \mu)) \mu$ 

■ Analysis of Strategies
■ Some Unmotivated do not Return

In[*]:= (* requires  $p < u_{low}$  *)

In[*]:= (* get solution conditional on targeting strategy *)

```

```

In[ ]:= tempsol = FullSimplify[Maximize[dfullrp * (p - c), p], assumptions]
Out[ ]:=  $\left\{ \frac{1}{4} (c - u + \Delta \mu (-1 + \alpha - \alpha \mu))^2, \left\{ p \rightarrow \frac{1}{2} (c + u + \Delta (1 + \alpha (-1 + \mu)) \mu) \right\} \right\}$ 

In[ ]:= (* Use solution to characterize profits, price, quantity, CW *)

In[ ]:= pfullrp = p /. tempsol[[2]];
proffullrp = tempsol[[1]];
qfullrp = FullSimplify[dfullrp /. p -> pfullrp];
cwfullrp = FullSimplify[exclcwrp + medcwrp + lowcwrp /. p -> pfullrp, assumptions];

In[ ]:= (* Get cutoff such that optimal price interior, i.e. p ≤ ulow *)

In[ ]:= FullSimplify[Solve[pfullrp == ulow, c]]
Out[ ]:=  $\left\{ \left\{ c \rightarrow u + \Delta \mu (-1 + \alpha - \alpha \mu) \right\} \right\}$ 

In[ ]:= cfullrphigh = FullSimplify[c /. Solve[pfullrp == ulow, c][[1]]];
    ■ Some Unmotivated Tender

In[ ]:= (* requires ulow < p < uunmotmot *)

In[ ]:= (* get solution conditional on targeting strategy *)

In[ ]:= tempsol = FullSimplify[Maximize[dmedrp * (p - c), p], assumptions]
Out[ ]:=  $\left\{ \frac{1}{4} (-c + u + \Delta + \alpha \Delta (-1 + \mu))^2 \mu, \left\{ p \rightarrow \frac{1}{2} (c + u + \Delta + \alpha \Delta (-1 + \mu)) \right\} \right\}$ 

In[ ]:= (* Use solution to characterize profits, price, quantity, CW *)

In[ ]:= pmedrp = p /. tempsol[[2]];
profmedrp = tempsol[[1]];
qmedrp = FullSimplify[dmedrp /. {p -> pmedrp}];
cwmedrp = FullSimplify[exclcwrp + medcwrp /. p -> pmedrp];

In[ ]:= (* Get cost thresholds such that optimum is interior *)

In[ ]:= Solve[pmedrp == ulow, c]
Out[ ]:=  $\left\{ \left\{ c \rightarrow u - \Delta + \alpha \Delta - \alpha \Delta \mu \right\} \right\}$ 

In[ ]:= Solve[pmedrp == uunmotmot, c]
Out[ ]:=  $\left\{ \left\{ c \rightarrow u + \Delta - \alpha \Delta - \alpha \Delta \mu \right\} \right\}$ 

In[ ]:= cmedrplow = c /. FullSimplify[Solve[pmedrp == ulow, c][[1]]];
cmedrphigh = c /. FullSimplify[Solve[pmedrp == uunmotmot, c][[1]]];
    ■ No unmotivated tender

In[ ]:= (* requires p > uunmotmot *)

In[ ]:= (* get solution conditional on targeting strategy *)

In[ ]:= tempsol = FullSimplify[Maximize[dexclrp * (p - c), p], assumptions]
Out[ ]:=  $\left\{ \frac{1}{4} (-c + u + \Delta)^2 \mu^2, \left\{ p \rightarrow \frac{1}{2} (c + u + \Delta) \right\} \right\}$ 

In[ ]:= (* Use solution to characterize profits, price, quantity, CW *)

```



```

In[ ]:= pexclrp = p /. tempsol[[2]];
profexclrp = tempsol[[1]];
qexclrp = FullSimplify[dexclrp /. p → pexclrp];
cwexclrp = FullSimplify[exclcwrp /. p → pexclrp];

In[ ]:= (* Get cost threshold such that optimum is interior *)

In[ ]:= Solve[pexclrp == uunmotmot, c]

Out[ ]:= {{c → u + Δ - 2 α Δ}}

In[ ]:= cexclrpflow = c /. FullSimplify[Solve[pexclrp == uunmotmot, c][[1]]];

■ Characterize Optimal Targeting

In[ ]:= (* For TeX: Print Results *)

In[ ]:= FullSimplify[{pfullrp, pmedrp, pexclrp}, assumptions]

Out[ ]:=  $\left\{ \frac{1}{2} (c + u + \Delta (1 + \alpha (-1 + \mu)) \mu), \frac{1}{2} (c + u + \Delta + \alpha \Delta (-1 + \mu)) \mu, \frac{1}{2} (c + u + \Delta) \right\}$ 

In[ ]:= FullSimplify[{qfullrp, qmedrp, qexclrp}, assumptions]

Out[ ]:=  $\left\{ \frac{1}{2} (-c + u + \Delta (1 + \alpha (-1 + \mu)) \mu), \frac{1}{2} (-c + u + \Delta + \alpha \Delta (-1 + \mu)) \mu, \frac{1}{2} (-c + u + \Delta) \mu^2 \right\}$ 

In[ ]:= FullSimplify[{proffullrp, profmedrp, profexclrp}, assumptions]

Out[ ]:=  $\left\{ \frac{1}{4} (c - u + \Delta \mu (-1 + \alpha - \alpha \mu))^2, \frac{1}{4} (-c + u + \Delta + \alpha \Delta (-1 + \mu))^2 \mu, \frac{1}{4} (-c + u + \Delta)^2 \mu^2 \right\}$ 

In[ ]:= FullSimplify[Collect[{cwfullrp, cwmedrp, cwexclrp}, Δ], assumptions]

Out[ ]:=  $\left\{ \frac{1}{8} (c^2 - 2 c u + u^2 + 2 (-c + u) \Delta \mu (1 + \alpha - \alpha \mu) + \Delta^2 \mu (4 - 3 \mu + \alpha (-1 + \mu) (-2 \mu + \alpha (4 + (-1 + \mu) \mu))) \right\},$ 
 $\frac{1}{8} \mu (c^2 + u^2 + 2 u \Delta (1 + \alpha - \alpha \mu) - 2 c (u + \Delta (1 + \alpha - \alpha \mu)) + \Delta^2 (1 + \alpha (-1 + \mu) (-2 + \alpha (3 + \mu)))) \right\},$ 
 $\frac{1}{8} (-c + u + \Delta)^2 \mu^2 \}$ 

■ Comparison 1: Full to medium

In[ ]:= FullSimplify[Solve[proffullrp == profmedrp, c], assumptions]

Out[ ]:=  $\left\{ \left\{ c \rightarrow u + \Delta (1 + \alpha (-1 + \mu)) \sqrt{\mu} \right\}, \left\{ c \rightarrow u + \Delta \sqrt{\mu} (-1 + \alpha - \alpha \mu) \right\} \right\}$ 

In[ ]:= c1temp = c /. FullSimplify[Solve[proffullrp == profmedrp, c][[1]]];
c2temp = c /. FullSimplify[Solve[proffullrp == profmedrp, c][[2]]];

In[ ]:= FullSimplify[c1temp > cfullrphigh, assumptions]

Out[ ]:= True

In[ ]:= FullSimplify[cfullrphigh > c2temp, assumptions]

Out[ ]:= True

In[ ]:= (* The relevant cutoff must be the lower root,
i.e. c2temp, which is always interior *)

In[ ]:= c1rp = FullSimplify[c2temp, assumptions];

```

### ■ Comparison 2: Medium to Exclusive

```

In[*]:= FullSimplify[Solve[profexclrp == profmedrp, c], assumptions]
Out[*]:=  $\left\{ \left\{ c \rightarrow u + \Delta + \alpha \Delta \left( -1 + \sqrt{\mu} \right) \right\}, \left\{ c \rightarrow u - \Delta \left( -1 + \alpha + \alpha \sqrt{\mu} \right) \right\} \right\}$ 

In[*]:= c1temp = c /. FullSimplify[Solve[profexclrp == profmedrp, c]] [[1]];
c2temp = c /. FullSimplify[Solve[profexclrp == profmedrp, c]] [[2]];

In[*]:= FullSimplify[c1temp > cmedrphigh > c2temp, assumptions]
Out[*]:= True

In[*]:= FullSimplify[c2temp > cmedrplow, assumptions]
Out[*]:=  $2 + \alpha \mu > \alpha \left( 2 + \sqrt{\mu} \right)$ 

In[*]:= (* The relevant cutoff must be the lower root, i.e. c2temp,
which is not always interior (may be too low) *)

In[*]:= c2rp = FullSimplify[c2temp, assumptions];

```

### ■ Comparison 3: Full to Exclusive

```

In[*]:= FullSimplify[Solve[profexclrp == proffullrp, c], assumptions]
Out[*]:=  $\left\{ \left\{ c \rightarrow u - \alpha \Delta \mu \right\}, \left\{ c \rightarrow u + \frac{\Delta \left( 2 + \alpha \left( -1 + \mu \right) \right) \mu}{1 + \mu} \right\} \right\}$ 

In[*]:= c1temp = c /. FullSimplify[Solve[profexclrp == proffullrp, c]] [[1]];
c2temp = c /. FullSimplify[Solve[profexclrp == proffullrp, c]] [[2]];

In[*]:= FullSimplify[c1temp < c2temp, assumptions]
Out[*]:= True

In[*]:= FullSimplify[c1temp > cexclrpplow, assumptions]
Out[*]:=  $1 + \alpha \mu < 2 \alpha$ 

In[*]:= (* The relevant cutoff must be the lower root, i.e. c1temp,
which is not always interior (may be too low) *)

In[*]:= c3rp = FullSimplify[c1temp, assumptions]
Out[*]:=  $u - \alpha \Delta \mu$ 

```

### ■ Relative Location of Thresholds

```

In[*]:= {c1rp, c2rp, c3rp}
Out[*]:=  $\left\{ u + \Delta \sqrt{\mu} \left( -1 + \alpha - \alpha \mu \right), u - \Delta \left( -1 + \alpha + \alpha \sqrt{\mu} \right), u - \alpha \Delta \mu \right\}$ 

In[*]:= (* Strategy: Establish that Ordering is fully determined
by comparison between  $\alpha$  and cutoff which we call alphacritrp *)

In[*]:= (* Step 1: get cutoff alphacritrp12 such that c1rp==c2rp *)

In[*]:= FullSimplify[Solve[c1rp == c2rp,  $\alpha$ ]]
Out[*]:=  $\left\{ \left\{ \alpha \rightarrow \frac{1}{1 + \sqrt{\mu} - \mu} \right\} \right\}$ 

```

```

In[*]:= alphacritrp12 =  $\alpha$  /. FullSimplify[Solve[c1rp == c2rp,  $\alpha$ ]] [[1]];
In[*]:= FullSimplify[1 > alphacritrp12 > 0, {1 >  $\mu$  > 0}]
Out[*]:= True

In[*]:= (* alphacritrp12 is interior in (0,1), denote it as alphacritrp *)
In[*]:= alphacritrp = alphacritrp12;
In[*]:= (* Step 2: get cutoff alphacritrp13 such that c1rp==c3rp *)
In[*]:= FullSimplify[Solve[c1rp == c3rp,  $\alpha$ ], {1 >  $\mu$  > 0}]
Out[*]:=  $\left\{ \left\{ \alpha \rightarrow \frac{1}{1 + \sqrt{\mu} - \mu} \right\} \right\}$ 

In[*]:= alphacritrp13 =  $\alpha$  /. FullSimplify[Solve[c1rp == c3rp,  $\alpha$ ], {1 >  $\mu$  > 0}] [[1]];
In[*]:= FullSimplify[alphacritrp == alphacritrp13]
Out[*]:= True

In[*]:= (* Step 3: get cutoff alphacritrp23 such that c2rp==c3rp *)
In[*]:= FullSimplify[Solve[c2rp == c3rp,  $\alpha$ ], {1 >  $\mu$  > 0}]
Out[*]:=  $\left\{ \left\{ \alpha \rightarrow \frac{1}{1 + \sqrt{\mu} - \mu} \right\} \right\}$ 

In[*]:= alphacritrp23 =  $\alpha$  /. FullSimplify[Solve[c2rp == c3rp,  $\alpha$ ], {1 >  $\mu$  > 0}] [[1]];
In[*]:= FullSimplify[alphacritrp23 == alphacritrp, {1 >  $\mu$  > 0}]
Out[*]:= True

In[*]:= (* This is the same threshold *)
In[*]:= assumptionsrphighalpha = {1 >  $\alpha$  > alphacritrp, 1 >  $\mu$  > 0, u > 0,  $\Delta$  > 0};
In[*]:= assumptionsrplowalpha = {alphacritrp >  $\alpha$  > 0, 1 >  $\mu$  > 0, u > 0,  $\Delta$  > 0};
In[*]:= FullSimplify[c1rp > c3rp > c2rp, {assumptionsrphighalpha}]
Out[*]:= True

In[*]:= FullSimplify[c1rp < c3rp < c2rp, assumptionsrplowalpha]
Out[*]:= True

In[*]:= (* check that pmedrp is indeed interior whenever
intermediate targeting occurs ( $\alpha$  < alphacritrp,  $c \in (c1rp, c2rp)$ ) *)
In[*]:= (* pmedrp > ulow for c = c1rp *)
In[*]:= FullSimplify[pmedrp > ulow /. c  $\rightarrow$  c1rp, assumptionsrplowalpha]
Out[*]:= True

In[*]:= (* pmedrp < uunmotmot for c = c2rp *)
In[*]:= FullSimplify[pmedrp < uunmotmot /. c  $\rightarrow$  c2rp, assumptionsrplowalpha]
Out[*]:= True

```

```

In[*]:= (* overall we have c3rp > c1rp if and only if  $\alpha < \text{alphacritrp}$  *)
(* So for low  $\alpha$  c1rp and c2rp determine
targeting behavior and intermediate targeting is possible,
for high  $\alpha$  it is not and c3rp determines targeting behavior *)

■  $\alpha$  uniquely determines targeting behavior

■ Characterization Complete

In[*]:= (* What can we say about possible combinations *)

■ 1.) Comparison CO vs BL

In[*]:= FullSimplify[ctildeco > ctilde, assumptions]

Out[*]:= True

In[*]:= (* cooling off period leads to weakly less exclusive targeting behavior *)

■ Possible Cases with CO

In[*]:= (* 1.) Cooling Off *)

In[*]:= (* For each of the possible combinations,
utilize what we know about c relative to thresholds
to assess directional change in quantities, price, and CW *)

In[*]:= (* A: Full to Full *)

In[*]:= FullSimplify[{Reduce[qfullco - qfull < 0], Reduce[pfullco - pfull < 0]}, assumptions]

Out[*]:= {True, True}

In[*]:= (* Consumer Welfare *)

In[*]:= cond1 =
  Simplify[Reduce[Simplify[Reduce[cwfullco > cwfull], assumptions]], (u |  $\alpha$  |  $\Delta$  |  $\mu$ )  $\in \mathbb{R}$ ]

Out[*]:=  $c < u + \Delta \mu - \frac{1}{2} \alpha \Delta (-4 + 7 \mu + \mu^2)$ 

In[*]:= FullSimplify[
  Equivalent[cond1, Simplify[Reduce[cwfullco > cwfull], assumptions]], assumptions]

Out[*]:= True

In[*]:= FullSimplify[cond1[[2]] > ctildeco, assumptions]

Out[*]:= True

In[*]:= (* Upper bound on c such that  $\Delta c > 0$  is above ctildeco;
c < ctildeco is required for full targeting in baseline *)

In[*]:= (* This implies that  $\Delta CW > 0$  for Full  $\rightarrow$  Full *)

In[*]:= (* B: Excl to Full *)

In[*]:= (* First establish  $\Delta q > 0$  and  $\Delta p < 0$  *)

In[*]:= FullSimplify[
  Reduce[FullSimplify[Reduce[{qfullco - qexcl > 0 && pfullco - pexcl < 0}], assumptions]],
  c < ctildeco]

Out[*]:= True

```

```

In[*]:= (* Consumer Welfare -- ΔCW *)

In[*]:= (* We establish that ΔCW>0 in two steps *)

In[*]:= (* For small μ it is straightforward to establish that cfullco > cexcl *)

In[*]:= (* Define critical μ *)

In[*]:= mucrit = 2 Sqrt[13] - 7;

In[*]:= FullSimplify[Simplify[
  Reduce[cfullco - cexcl > 0 && 0 < μ < mucrit && assumptions], μ < mucrit], assumptions]

Out[*]:= True

In[*]:= (* For large μ this becomes slightly more involved *)

In[*]:= Reduce[cfullco > cexcl && 1 > μ > mucrit && assumptions]

Out[*]:=  $-7 + 2\sqrt{13} < \mu < 1 \&\& 0 < \alpha < 1 \&\& \Delta > 0 \&\& u > 0 \&\&$ 

$$\left( c < u + \Delta \mu + 2\alpha \Delta \mu - \sqrt{-3\alpha^2 \Delta^2 \mu + 14\alpha^2 \Delta^2 \mu^2 + \alpha^2 \Delta^2 \mu^3} \mid \mid \right.$$


$$\left. c > u + \Delta \mu + 2\alpha \Delta \mu + \sqrt{-3\alpha^2 \Delta^2 \mu + 14\alpha^2 \Delta^2 \mu^2 + \alpha^2 \Delta^2 \mu^3} \right)$$


In[*]:= temp = Simplify[Simplify[
  Reduce[cfullco > cexcl && 1 > μ > mucrit && assumptions], μ > mucrit], assumptions]

Out[*]:=  $c < u + \Delta \left( \mu + 2\alpha \mu - \alpha \sqrt{\mu(-3 + 14\mu + \mu^2)} \right) \mid \mid c > u + \Delta \left( \mu + 2\alpha \mu + \alpha \sqrt{\mu(-3 + 14\mu + \mu^2)} \right)$ 

In[*]:= (* We focus on the first inequality and denote by ctemp
  the upper bound on c which ensures that cfullco > cexcl *)

In[*]:= ctemp = temp[[1]][[2]]

Out[*]:=  $u + \Delta \left( \mu + 2\alpha \mu - \alpha \sqrt{\mu(-3 + 14\mu + \mu^2)} \right)$ 

In[*]:= (* If c < ctemp and μ > mucrit we have ΔCW > 0 *)

In[*]:= Simplify[Simplify[Simplify[Reduce[cfullco > cexcl && assumptions],
  {c < ctemp, μ > mucrit}], assumptions], c < ctemp]

Out[*]:= True

In[*]:= (* As full targeting materializes post policy adoption we know that c < ctildeco *)

In[*]:= (* It hence suffices for ΔCW > 0 to show that ctildeco < ctemp *)

In[*]:= Simplify[Simplify[Reduce[ctildeco < ctemp], μ > mucrit], assumptions]

Out[*]:= True

In[*]:= (* Overall we hence have ΔCW > 0 for Excl → Full *)

In[*]:= (* C: Excl to Excl *)

In[*]:= (* First establish Δq < 0 and Δp = 0 *)

In[*]:= FullSimplify[Reduce[qexclco - qexcl < 0 && pexclco - pexcl == 0],
  assumptions && c > ctildeco && c < expmot]

Out[*]:= True

In[*]:= (* ΔCW is ambiguous *)

```

In[\*]:= **t = Simplify[Simplify[Reduce[cwexclco - cwexcl > 0 && assumptions], assumptions],  
{c > ctildeco, c < expmot}]**

Out[\*]=  $c > u + \Delta \left( 3\alpha \left( -1 + \mu \right) + \mu \right)$

In[\*]:= **(\* Define the critical c as ctemp \*)**

In[\*]:= **ctemp = t[[2]]**

Out[\*]=  $u + \Delta \left( 3\alpha \left( -1 + \mu \right) + \mu \right)$

In[\*]:= **Simplify[ctildeco < ctemp < expmot, assumptions]**

Out[\*]=  $5\mu > 3$

In[\*]:= **(\* For  $\mu > 3/5$ , there are constellations where Excl  $\rightarrow$  Excl reduces welfare,  
specifically for  $c > ctemp$  with  $ctemp \in (ctildeco, expmot)$  \*)**

■ 2.) Comparison BL to RP

In[\*]:= **FullSimplify[c1rp < ctilde, {1 >  $\mu$  > 0,  $\Delta$  > 0, 0 <  $\alpha$  < alphacritrp}]**

Out[\*]=  $1 + \alpha \sqrt{\mu} > 2\alpha$

In[\*]:= **FullSimplify[c2rp > ctilde, {1 >  $\mu$  > 0,  $\Delta$  > 0, 1 >  $\alpha$  > 0}]**

Out[\*]= True

In[\*]:= **FullSimplify[c3rp > ctilde, {1 >  $\mu$  > 0,  $\Delta$  > 0, 1 >  $\alpha$  > alphacritrp}]**

Out[\*]= True

In[\*]:= **(\* Know that Exclusive targeting materializes for larger cutoff due to c2rp >  
ctilde and exclusive targeting iff  $c > \text{Max}[c2rp, c3rp]$  \*)**

In[\*]:= **(\* So Full  $\rightarrow$  Excl is ruled out \*)**

In[\*]:= **(\* Get  $\Delta p$ ,  $\Delta q$ ,  
 $\Delta CW$  for the 5 possible combinations of pre and post policy targeting behavior \*)**

In[\*]:= **(\* A: Full to Full \*)**

In[\*]:= **(\* Establish  $\Delta q < 0$ ,  $\Delta p < 0$  \*)**

In[\*]:= **FullSimplify[qfullrp - qfull < 0 && pfullrp - pfull < 0, assumptions]**

Out[\*]= True

In[\*]:= **(\* Establish  $\Delta CW > 0$  \*)**

In[\*]:= **(\* Get cutoff ctemp such that this holds for  $c < ctemp$  \*)**

In[\*]:= **t = FullSimplify[Reduce[cwfullrp > cwfull && assumptions], assumptions]**

Out[\*]=  $c < u + \frac{2\Delta}{\alpha} + \frac{1}{2}\Delta\mu \left( 2 + \alpha - \alpha\mu \right)$

In[\*]:= **ctemp = t[[2]]**

Out[\*]=  $u + \frac{2\Delta}{\alpha} + \frac{1}{2}\Delta\mu \left( 2 + \alpha - \alpha\mu \right)$

In[\*]:= **(\* ctemp > ctilde is sufficient for  $\Delta CW > 0$  as  $c < ctilde <$   
ctemp follows due to full market coverage pre intervention \*)**

```
In[*]:= FullSimplify[ctemp > ctilde, assumptions]
```

```
Out[*]:= True
```

```
In[*]:= (* B: Full to Int *)
```

```
(* Exploit that Intermediate targeting with RP only possible for  $\alpha <$   
alphacritrp as this is necessary for  $c1rp < c3rp < c2rp$  *)
```

```
In[*]:= (* Full to Int requires that  $c1rp < c < c2rp$  but  $c < ctilde$ ,  
so  $ctilde > c1rp$  is necessary *)
```

```
(* This implies a tighter bound on  $\alpha$ , namely  $0 < \alpha < fullmedalpha < alphacritrp$  *)
```

```
In[*]:= FullSimplify[Reduce[c1rp < ctilde && assumptions], assumptions]
```

```
Out[*]:=  $2\alpha \leq 1 \mid \mid \alpha(4 + \alpha(-4 + \mu)) > 1$ 
```

```
(* we can ignore the  $\alpha \leq 1/2$  as the constraint derived below subsumes this *)
```

```
In[*]:= Simplify[Solve[c1rp == ctilde,  $\alpha$ ], assumptions]
```

```
Out[*]:=  $\left\{ \left\{ \alpha \rightarrow \frac{1}{2 - \sqrt{\mu}} \right\} \right\}$ 
```

```
In[*]:= fullmedalpha =  $\alpha$  /. Simplify[Solve[c1rp == ctilde,  $\alpha$ ], assumptions][[1]]
```

```
Out[*]:=  $\frac{1}{2 - \sqrt{\mu}}$ 
```

```
In[*]:= Simplify[ $0 < 1/2 < fullmedalpha < alphacritrp, 1 > \mu > 0$ ]
```

```
Out[*]:= True
```

```
In[*]:= (* fullmedalpha > 1/2 ensures that this is indeed the more relevant constraint *)
```

```
In[*]:= FullSimplify[ctilde > c1rp && assumptions && fullmedalpha >  $\alpha$ ,  
assumptions && fullmedalpha >  $\alpha$ ]
```

```
Out[*]:= True
```

```
In[*]:= (* Having established this, we now consider  $\Delta p$ ,  $\Delta q$  and  $\Delta CW$  *)
```

```
In[*]:= (* First establish  $\Delta p > 0$  *)
```

```
In[*]:= FullSimplify[pmedrp > pfull, assumptions]
```

```
Out[*]:= True
```

```
In[*]:= (* Next consider  $\Delta q$  *)
```

```
In[*]:= t = Simplify[Reduce[qmedrp < qfull && assumptions], assumptions]
```

```
Out[*]:=  $c < u + \alpha \Delta \mu$ 
```

```
In[*]:= ctemp = t[[2]]
```

```
Out[*]:=  $u + \alpha \Delta \mu$ 
```

```
In[*]:= (* qmedrp < qfull for  $c < ctemp$  *)
```

```
In[*]:= condtemp = FullSimplify[Reduce[ctilde > ctemp > c1rp], assumptions &&  $0 < \alpha < fullmedalpha$ ]
```

```
Out[*]:=  $\mu > \frac{\alpha^2}{(1 - 2\alpha)^2}$ 
```

```

In[*]:= FullSimplify[Exists[ $\alpha$ ,  $0 < \alpha < \text{fullmed}\alpha$  && assumptions], condtemp], assumptions]
Out[*]:= True

(* Clearly the above can be satisfied so that  $\Delta q > 0$  and  $\Delta q < 0$  is possible for  $c < \text{ctemp}$  and  $c > \text{ctemp}$ , respectively;  $\Delta q$  is hence ambiguous *)

In[*]:= (* Next consider  $\Delta CW$  *)

In[*]:= cond = FullSimplify[Reduce[cwmedrp > cwfull && assumptions], assumptions]
Out[*]:=  $u < c + (1 + \alpha) \Delta \sqrt{\mu} + \alpha \Delta \mu$  &&  $c + \alpha \Delta \mu < u + (1 + \alpha) \Delta \sqrt{\mu}$ 

In[*]:= (* Define c1temp, c2temp such that  $\Delta CW > 0$  iff  $c \in (c1temp, c2temp)$  *)

In[*]:= t = FullSimplify[Solve[cwmedrp == cwfull, c], assumptions]
Out[*]:=  $\{ \{ c \rightarrow u - (1 + \alpha) \Delta \sqrt{\mu} - \alpha \Delta \mu \}, \{ c \rightarrow u + (1 + \alpha) \Delta \sqrt{\mu} - \alpha \Delta \mu \} \}$ 

In[*]:= c1temp = c /. t[[1]]; c2temp = c /. t[[2]];

In[*]:= FullSimplify[c1temp < c2temp, assumptions]
Out[*]:= True

In[*]:= FullSimplify[Equivalent[cond, c1temp < c && c < c2temp], assumptions]
Out[*]:= True

In[*]:= (* Recall that we know that  $c \in (c1rp, ctilde)$  for this targeting behavior combination *)

In[*]:= (* It is hence sufficient to establish  $c1rp > c1temp$  and  $ctilde < c2temp$  for  $\Delta CW > 0$  *)

In[*]:= FullSimplify[c1rp > c1temp && ctilde < c2temp, assumptions]
Out[*]:= True

In[*]:= (* This implies  $\Delta CW > 0$  *)

In[*]:= (* D: Excl to Full *)

In[*]:= (* For this we require that  $c > ctilde$  and  $c < \text{Min}[c1rp, c3rp]$  *)

In[*]:= (* First establish  $\Delta q > 0$  *)

In[*]:= Simplify[Reduce[qfullrp > qexcl && assumptions], assumptions]
Out[*]:=  $c + (-1 + 2\alpha) \Delta \mu < u$ 

In[*]:= (* define critical ctemp such that  $\Delta q > 0$  iff  $c < \text{ctemp}$  *)

In[*]:= t = Solve[qfullrp == qexcl, c]
Out[*]:=  $\{ \{ c \rightarrow u + \Delta \mu - 2\alpha \Delta \mu \} \}$ 

In[*]:= ctemp = c /. t[[1]]
Out[*]:=  $u + \Delta \mu - 2\alpha \Delta \mu$ 

In[*]:= (*  $c < \text{ctemp}$  is sufficient for  $\Delta q > 0$ . In turn,  $\text{ctemp} > c3rp$  is sufficient for this as  $c < c3rp$  is necessary for full targeting post intervention *)

```



```
In[*]:= FullSimplify[ctemp > c3rp, assumptions]
```

```
Out[*]:= True
```

```
In[*]:= (* We hence have  $\Delta q > 0$  *)
```

```
In[*]:= (* Next establish  $\Delta p < 0$  *)
```

```
In[*]:= FullSimplify[pfullrp < pexcl, assumptions]
```

```
Out[*]:= True
```

```
In[*]:= (* Next establish  $\Delta CW > 0$  *)
```

```
In[*]:= (* We establish  $\Delta CW > 0$  by (i) showing that  $\Delta CW > 0$  at  $c \rightarrow c1rp$  and  $c \rightarrow c3rp$ ,  
and (ii) that  $\Delta CW$  is decreasing in  $c$  for  $c < c1rp$  and  $c < c3rp$  *)
```

```
In[*]:= (* This implies that irrespective of whether  $c1rp$  or  $c3rp$  is the threshold  
for full targeting post policy adoption,  $\Delta CW$  is strictly positive as  $c \in$   
( $c_{\text{tilde}}, \text{Min}[c1rp, c3rp]$ ) is required for  $\text{Excl} \rightarrow \text{Full}$  *)
```

```
In[*]:= (* Define  $\Delta CW = \text{deltaCW}$  *)
```

```
In[*]:= deltaCW = FullSimplify[cwfullrp - cwexcl, assumptions]
```

```
Out[*]:= 
$$-\frac{1}{8}(-1 + \mu) \left( c^2 + u^2 + 2u(1 + 2\alpha)\Delta\mu - 2c(u + (1 + 2\alpha)\Delta\mu) + \Delta^2\mu(4 + \mu + 4\alpha\mu - \alpha^2(1 + \mu)^2) \right)$$

```

```
In[*]:= (* Show that  $\Delta CW > 0$  at  $c \rightarrow c1rp$  *)
```

```
In[*]:= FullSimplify[FullSimplify[deltaCW /. c -> c1rp, assumptions] > 0, assumptions]
```

```
Out[*]:= True
```

```
In[*]:= (* Show that  $\Delta CW > 0$  at  $c \rightarrow c3rp$  *)
```

```
In[*]:= FullSimplify[FullSimplify[deltaCW /. c -> c3rp, assumptions] > 0, assumptions]
```

```
Out[*]:= True
```

```
(* Get threshold ctemp such that  $\Delta CW$  is decreasing in  $c$  iff  $c <$   
ctemp:  $D[\Delta CW, c] < 0$  for  $c < \text{ctemp}$  *)
```

```
In[*]:= t = FullSimplify[Reduce[D[deltaCW, c] < 0 && assumptions], assumptions]
```

```
Out[*]:=  $c < u + \Delta\mu + 2\alpha\Delta\mu$ 
```

```
In[*]:= ctemp = t[[2]]
```

```
Out[*]:=  $u + \Delta\mu + 2\alpha\Delta\mu$ 
```

```
In[*]:= FullSimplify[D[deltaCW, c] < 0, {c < ctemp, 1 >  $\mu > 0$ , 1 >  $\alpha > 0$ , u > 0}]
```

```
Out[*]:= True
```

```
In[*]:= (* Show that  $\text{ctemp} > c1rp$  and  $\text{ctemp} > c3rp$  *)
```

```
In[*]:= FullSimplify[ctemp > c1rp && ctemp > c3rp, assumptions]
```

```
Out[*]:= True
```

```
In[*]:= (* Overall we hence have  $\Delta CW > 0$  because  $\text{deltaCW}$  is decreasing in  $c$  up to  $c1rp$   
or  $c3rp$  which are the respective upper bound for the targeting region *)
```

```
In[*]:= (* H: Excl to Med *)
```

```

In[*]:= (* Note that Excl → Int implies multiple things *)

In[*]:= (* First, it needs to be that c1rp < c3rp < c2rp, i.e. α < alphacritrp,
so that Intermediate targeting post policy can materialize *)

In[*]:= (* Second, for Excl → Int we require that c ∈ (c1rp, c2rp) and c > ctilde *)

In[*]:= (* As ctilde < c2rp holds generically,
there always is such a parameter constellation provided that α < alphacritrp *)

In[*]:= (* Note that alphacritrp > 1/2 *)

In[*]:= FullSimplify[alphacritrp > 1/2, 1 > μ > 0]

Out[*]:= True

In[*]:= (* On Δq: We show Δq > 0 iff α < 1/2;
α can lie above or below 1/2 as it is only required to be below alphacritrp *)

In[*]:= Simplify[Reduce[qmedrp > qexcl && assumptions], assumptions]

Out[*]:= 2 α < 1

In[*]:= (* On Δp: We show Δp > 0 iff α < 1/2;
α can lie above or below 1/2 as it is only required to be below alphacritrp *)

In[*]:= Simplify[Reduce[pmedrp > pexcl && assumptions], assumptions]

Out[*]:= 2 α < 1

In[*]:= (* Finally we need to consider ΔCW *)

In[*]:= deltaCW = FullSimplify[cwmedrp - cwexcl, assumptions]

Out[*]:=  $\frac{1}{8} \Delta (-1 + \mu) \mu \left( (1 + 2\alpha) (2c - 2u - \Delta) + (-1 - 2\alpha + 4\alpha^2) \Delta \mu \right)$ 

In[*]:= (* Define ctemp such that ΔCW > 0 iff c < ctemp *)

In[*]:= t = Simplify[Reduce[deltaCW > 0 && assumptions],
assumptions && c > ctilde && c < c2rp && 0 < α < alphacritrp]

Out[*]:=  $c < u + \frac{\Delta (1 + \mu - 4\alpha^2 \mu + 2\alpha (1 + \mu))}{2 + 4\alpha}$ 

In[*]:= ctemp = t[[2]]

Out[*]:=  $u + \frac{\Delta (1 + \mu - 4\alpha^2 \mu + 2\alpha (1 + \mu))}{2 + 4\alpha}$ 

In[*]:= (* ctemp lies strictly above ctilde, so ΔCW > 0 for c → ctilde *)

In[*]:= Simplify[Reduce[ctemp > ctilde], assumptions]

Out[*]:= True

In[*]:= Simplify[Reduce[(deltaCW /. c → ctilde) > 0 && assumptions], assumptions]

Out[*]:= True

In[*]:= (* If we could also establish that ctemp > c2rp then by c <
c2rp for Intermediate targeting post adoption we know that c <
ctemp and hence ΔCW > 0 *)

```

*In[\*]:=* (\* In contrast, if ctemp < c2rp then there exists a cost range,  
specifically (ctemp,c2rp), such that  $\Delta CW < 0$  \*)

*In[\*]:=* (\* define deltac = ctemp - c2rp \*)

*In[\*]:=* deltac = FullSimplify[ctemp - c2rp, assumptions]

$$\text{Out[*]} = \Delta \left( -1 + \alpha + \alpha \sqrt{\mu} + \frac{1 + \mu + 2\alpha(1 + \mu - 2\alpha\mu)}{2 + 4\alpha} \right)$$

*In[\*]:=* Simplify[Reduce[deltac < 0 && assumptions &&  $0 < \alpha < \text{alphacritrp}$ ],  
assumptions &&  $0 < \alpha < \text{alphacritrp}$ ]

$$\text{Out[*]} = 2\alpha < 1 \&\& 2 \sqrt{\frac{\alpha^2 (1 + 2\alpha)^3 (1 - 7\alpha^2 + 10\alpha^3)}{(1 + 2\alpha - 4\alpha^2)^4}} + \mu < \frac{1 + 2\alpha - 6\alpha^2 + 24\alpha^4}{(1 + 2\alpha - 4\alpha^2)^2}$$

*In[\*]:=* (\* For  $\alpha > 1/2$ , we hence have deltac > 0 and therefore ctemp >  
c2rp which implies that  $c < c2rp < ctemp$  and thus  $\Delta CW > 0$  \*)

*In[\*]:=* FullSimplify[Reduce[ctemp > c2rp && assumptions &&  $1/2 < \alpha$ ],  
assumptions &&  $1/2 < \alpha < \text{alphacritrp}$ ]

*Out[\*]=* True

*In[\*]:=* FullSimplify[Reduce[deltaCW > 0 &&  $c < ctemp$  && assumptions], assumptions &&  $c < ctemp$ ]

*Out[\*]=* True

*In[\*]:=* (\* Restrict attention to the case where  $\alpha < 1/2$  \*)

*In[\*]:=* t = FullSimplify[Solve[FullSimplify[Reduce[deltac == 0], assumptions],  $\alpha$ ], assumptions]

$$\text{Out[*]} = \left\{ \left\{ \alpha \rightarrow \frac{\sqrt{\mu} + \mu - \sqrt{(1 + \sqrt{\mu})(4 - 7\mu + 5\mu^{3/2})}}{4(-1 - \sqrt{\mu} + \mu)} \right\} \right\}$$

*In[\*]:=* exclintcriticalalpha =  $\alpha /. t[[1]]$

$$\text{Out[*]} = \frac{\sqrt{\mu} + \mu - \sqrt{(1 + \sqrt{\mu})(4 - 7\mu + 5\mu^{3/2})}}{4(-1 - \sqrt{\mu} + \mu)}$$

*In[\*]:=* FullSimplify[alphacritrp > exclintcriticalalpha > 0,  $1 > \mu > 0$ ]

*Out[\*]=* True

*In[\*]:=* FullSimplify[exclintcriticalalpha <  $1/2$ ,  $1 > \mu > 0$ ]

*Out[\*]=* True

(\* For  $\alpha < \text{exclintcriticalalpha}$  there exists a cost range  $c \in$   
(ctemp,c2rp) such that Excl  $\rightarrow$  Int targeting materializes with  $\Delta CW < 0$  \*)

*In[\*]:=* FullSimplify[c1rp < ctemp && c1rp < ctilde && ctilde < ctemp && ctemp < c2rp,  
{ $0 < \alpha < \text{exclintcriticalalpha}$ ,  $1 > \mu > 0$ ,  $\Delta > 0$ }]

*Out[\*]=* True

(\* Verify that cwexcl >  
0 indeed holds (this is necessary as  $CW > 0$  always holds with a RP) \*)

```
In[*]:= FullSimplify[cwexcl > 0 /. c → ctemp, assumptions && 0 < α < exclintcriticalalpha]
```

```
Out[*]:= True
```

```
In[*]:= (* For α > exclintcriticalalpha, ΔCW > 0 holds generically for Excl → Int *)
```

```
In[*]:= (* As exclintcriticalalpha < 1/2,
      this subsumes α > 1/2 which we had already established *)
```

```
In[*]:= (* I: Excl to Excl *)
```

```
In[*]:= (* First adress Δq *)
```

```
In[*]:= (* Define ctemp such that Δq > 0 iff c > ctemp *)
```

```
In[*]:= t = Simplify[Reduce[qexclrp > qexcl, c], assumptions]
```

```
Out[*]:= c > u + α Δ
```

```
In[*]:= ctemp = t[[2]]
```

```
Out[*]:= u + α Δ
```

```
In[*]:= (* For Excl →
      Excl we require that c > Max[c2rp,c3rp] as c2rp > ctilde holds generically *)
```

```
In[*]:= FullSimplify[ctemp < c2rp || ctemp < c3rp, assumptions]
```

```
Out[*]:= α (2 + √μ) < 1
```

```
In[*]:= (* For small α we hence have ctemp < Max[c2rp,c3rp] and thus that Δq > 0 *)
```

```
In[*]:= (* For large α in contrast Δq < 0 is possible *)
```

```
In[*]:= (* Overall, Δq is ambiguous *)
```

```
In[*]:= (* Next show Δp > 0 *)
```

```
In[*]:= Simplify[Reduce[pexclrp > pexcl && assumptions], assumptions]
```

```
Out[*]:= True
```

```
In[*]:= (* Finally analyze ΔCW *)
```

```
In[*]:= cond1 = Simplify[Reduce[cwexclrp > cwexcl, c], assumptions]
```

```
Out[*]:= u < c + Δ (α + √α² (4 - 3 μ) + μ + 2 α μ) && c + α Δ < u + Δ √α² (4 - 3 μ) + μ + 2 α μ
```

```
In[*]:= (* Above shows that ΔCW > 0 iff c ∈
      (c1temp,c2temp) with c1temp and c2temp characterized as follows *)
```

```
In[*]:= t = FullSimplify[Solve[cwexclrp == cwexcl, c], assumptions]
```

```
Out[*]:= { {c → u + Δ (-α + √μ + α (4 α + 2 μ - 3 α μ)) }, {c → u - Δ (α + √μ + α (4 α + 2 μ - 3 α μ)) } }
```

```
In[*]:= c1temp = c /. t[[1]]; c2temp = c /. t[[2]];
```

```
In[*]:= FullSimplify[c1temp > c2temp, assumptions]
```

```
Out[*]:= True
```

```
(* Verify that condition on cost is equivalent to ΔCW > 0 *)
```

```
In[*]:= cond = Simplify[Reduce[cwexclrp > cwexcl && assumptions], assumptions];
```

```
In[ ]:= FullSimplify[
  Reduce[Equivalent[cond, c2temp < c && c < c1temp] && assumptions], assumptions]
```

```
Out[ ]:= True
```

```
(* Note that c1temp > expmot and hence that for  $\Delta CW > 0$  it is sufficient that c2temp < c2rp as c > c2rp is necessary for Exclusive Targeting post intervention *)
```

```
In[ ]:= FullSimplify[c1temp > expmot, assumptions]
```

```
Out[ ]:= True
```

```
In[ ]:= FullSimplify[c2temp < ctilde, assumptions]
```

```
Out[ ]:= True
```

```
In[ ]:= (* Hence, we necessarily have  $\Delta CW > 0$  *)
```